


Automatische spoel installatie voor legionella preventie.

NTKC - Bennie ten Haken

12 – Aug. - 2017



Automatische spoel installatie voor legionella preventie.

Bennie ten Haken, 12 – Aug - 2017

Deze installatie is aangelegd, getest en gebruikt op de NTKC-kampeerterrinen Masterveld. Na de eerste test periode in 2016, zijn er in 2017 in totaal 17 exemplaren van deze automaat gemaakt. Deze zullen vervolgens op in totaal 15 NTKC kampeerterrinen geïnstalleerd gaan worden. Op het ontwerp van deze installatie rust geen enkel eigendomsrecht. Alle mogelijke partijen worden van harte uitgenodigd deze installatie te kopiëren en/of te verbeteren

Automatische spoel installatie voor legionella preventie.

Bennie ten Haken, 26-Jan-2016

Bedieningskastje

Bediening: Aan/uit met stekker op 230 V ingang

Ingang: 1^e thermometer in het bedieningskastje

2^e thermometer extern (bij de warmste waterplek in het gebouw)

Uitgangen: 6x 24V AC uitgang via relais 2 met draadjes naar magneetventiel (maak/breek)

Display: Actuele Temperaturen

Actuele spoel status (1 t/m 6)

Verstreken tijd sinds laatste spoelbeurt

Maximale temperatuur sinds laatste spoelbeurt

Bediening:

- Handmatig spoelen: 1x Uit & Aan (=reset)
- Temperatuur aflezen: klopje op de kast → display aan
- IJken: vergelijk display met andere geijkte thermometer
- Water afsluiten: Alle kranen open en dan Uit & Aan, 5-15 min wachten, Uit

Programma inhoud:

Start na power-up (=reset)

1) Direct Spoelen

- Alle aangesloten kanalen voor 2 minuten spoelen (1-6, nummer op display)
- Tussendoor 1 sec. pauze
- Max-T = 0
- Dagteller = 0

2) T-meting in loop

- T direct op display
- Max-T = T of Max-T (vgl. intern & extern)
- Geluid meten via circuit aangesloten op digital in 1 → licht aan schakelen
- Tel de uren en als $T < 24$ → ga terug naar begin stap 2.

3) Na 24 hr. Dagteller = Dagteller+1

4) Bij Max-T>24 of Dagteller > 6

- Complete reset v.d. Arduino
- Dan starten we weer met spoelprocedure

5) Terug naar stap 2)

Aansluitingen

Relais-sturing 1-6 – Digital out pin 8-13 (VMA05 – Relais 1-6)

Relais Stroommeting – Analog Pin 0 (VMA05 - Analog in 1)

T-intern – Analog Pin 1 (VMA05 - Analog in 2)

T-extern – Analog Pin 2 (VMA05 - Analog in 3) – Losse draad naar buiten

Display SDA - Analog Pin 4 (VMA05 pin 5 ombuigen of afknippen)

Display SCL - Analog pin 5 (VMA05 pin 6 ombuigen of afknippen)


Display GND - GND

Display VCC - 5V

Geluidsensor – Digital-In pin nr. 2 (VMA05 - Dig in 1)

Elektrische aansluitingen op de Velleman VMA05 I/O shield

Bennie ten Haken 25 – Mei - 2017



Aansluitingen:

In de 230 aansluitingen zitten 2-zekeringen van 0.5A (1=reserve)


24 VAC aansluiting (=naar kraan-relais): Luidspreker snoer 2x 0.5mm²: Praxis 50m → 18.30 Euro

Thermometer: 3 (of 4) ader telefoon snoer → kies een makkelijk monteerbare soort met voldoende koper die goed in de kroonsteentjes kan en bvk ook te solderen is.



Mechanische onderdelen: Perspexplaatje en Alu-steuntje (2x)

Bennie ten Haken 23-Nov.-2016


Boormal perspex-plaatje (gaten 3&5mm)




Aluminium steuntje (2x10mm, 2x)



Hardware



<http://ledsee.com/index.php/arduino-modules2013-02-16-10-32-17/arduino-main-boards/arduino-uno-2012-r3-detail> 15 Euro → UT-Studenten club Scintilla **12 Euro**



<https://www.conrad.nl/nl/arduino-io-shield-velleman-vma05-module-678492.html> Conrad **39,99 Euro**
<http://www.electroshopdendermonde.be/index.php/webshop/electroshop/arduino-shields/io-shield-voor-arduino/> per 10 → 28,19 Euro
<https://sinuss.nl/product/1308/vma05-i-o-shield-voor-arduino-> per 10 --> 32,26 Euro



<http://www.hobbyelectronica.nl/product/2004-lcd-i2c-blauw-backlight/> Hobbyelectronica **13,95 Euro**



www.scintilla.utwente.nl/nl/stores **1.40 Euro** → Studenten club Scintilla



<http://www.hobbyelectronica.nl/product/geluid-detectie-sensor-tk0862/>
Geluid detectie sensor TK0862 **2,95 Euro**



<https://www.conrad.nl/nl/transformatoren-voor-din-railmontage-3-m-te-24-vac-20-w-comatec-inhoud-1-stuks-520393.html>
Transformatoren voor DIN-railmontage 3 M (TE) 24 V/AC 20 W Comatec **24,- Euro**



<https://www.conrad.nl/nl/mean-well-mdr-10-12-din-rail-netvoeding-12-vdc-084-a-10-w-1-x-1297344.html> Mean Well MDR-10-12 Din-rail netvoeding 12 V/DC 0.84 A 10 W 1 x. **16 Euro & Koning&Hartmann 11,80 Euro**



<https://www.conrad.nl/nl/fibox-abs-mh-125-t-installatiebehuizing-ip66ip67-230-x-140-x-125-abs-lichtgrijs-ral-7035-1-stuks-536014.html> 536014 Behuizing ABS Piccolo ABS MH 125, Fibox ABS MH 125 T
Installatiebehuizing IP66/IP67 230 x 140 x 125 ABS Lichtgrijs (RAL 7035) **31,09 Euro 10x 28,60 Euro**



<https://www.conrad.nl/nl/hellermannityton-delta-3fbv-geperforeerd-plaatstaal-1-stuks-545595.html>
HellermannTyton DELTA-3F/BV Geperforeerd Plaatstaal **11,06 Euro** → dat is voor ca. 10 stuks

Componenten & aansluitingen:

<https://www.conrad.nl/nl/apparaatstekker-c14-stekker-inbouw-verticaal-totaal-aantal-polen-3-10-a-zwart-1-stuks-717133.html>
c14 Stekker, inbouw verticaal Totaal aantal polen: 3 10 A Zwart – **2,80 Euro**

<https://www.conrad.nl/nl/eska-515314-buiszekering-x-l-5-mm-x-15-mm-05-a-250-v-traag-t-inhoud-10-stuks-536409.html>
ESKA 515.314 Buiszekering (Ø x l) 5 mm x 15 mm 0.5 A 250 V Traag -T- Inhoud 10 stuks – **3,00 Euro**

<https://www.conrad.nl/nl/bkl-electronic-072053-laagspannings-aansluitkabel-laagspanningsstekker-open-einde-55-mm-2-m-1-stuks-1371055.html>
Aansluiting: BKL Electronic 072053 Laagspannings-aansluitkabel Laagspanningsstekker – **2.60 Euro**

<https://www.conrad.nl/nl/digitus-stroom-aansluitkabel-1x-geaarde-stekker-1x-apparaatstekker-female-c13-10a-180-m-zwart-678273.html> Digitus Stroom Aansluitkabel - **2,80 Euro**

<https://www.conrad.nl/nl/kroonsteen-flexibel-05-15-mm-massief-05-15-mm-aantal-polen-12-10-stuks-wit-523690.html>
10 Stuks 12-Kroonsteentjes voor 8.08 Euro voor 2 - stuks - **1.62 Euro**

Schijfvaristor S07K300 470 V Epcos S07K300 - **0.30 Euro**

Diode 1N4007 1000V/1A Diotec 1N4007 Soort behuizing DO-41 I(F) 1 A Blokkeerspanning U(R) 1 – **0.08 Euro**

MFR1145 Metaalfilmweerstand 1 kΩ Axiaal bedraad 0414 1 W - **0.15 Euro**

Elektrolytische condensator 22 µF 16 V/DC 20 % (Ø x h) 5 mm x 11 mm ECR1CPT220MFF200511 - **0.08 Euro**

```

// LegioSpoel programma in C++, Bennie ten Haken,16-Apr-2016

#include <Wire.h> // Comes with Arduino IDE

    boolean simulatieAan = false; // Simulatie uit
//    boolean simulatieAan = true; // Simulatie aan --> 1000x versnelt, 1 uur = 3.6 Sec.

// Aansluitingen:
// Arduino digitaal-uit 8 t/m 13, Relais 1 t/m 6 van de Velleman VMA05
// Stroommetingen op analoog-in 0      (=VMA05 Analog-In 1)
// T-intern op analoog-in 1            (=VMA05 Analog-In 2)
// T-intern op analoog-in 2            (=VMA05 Analog-In 3)
// Vrijgehouden analoog-in 3          (=VMA05 Analog-In 4)
// Digitaal display op analoog-in 4&5 (=VMA05 Analog-In 5&6)
// Geluid detectie digitaal in 2      (=VMA05 Digitaal in 1)

// Get the LCD I2C Library here:
// https://bitbucket.org/fmalpartida/new-liquidcrystal/downloads
// Move any other LCD libraries to another folder or delete them
// See Library "Docs" folder for possible commands etc.
#include <LiquidCrystal_I2C.h>

// set the LCD address to 0x27 for a 20 chars 4 line display
// Set the pins on the I2C chip used for LCD connections:
//          addr, en,rw,rs,d4,d5,d6,d7,b1,blpol
LiquidCrystal_I2C lcd(0x27, 2, 1, 0, 4, 5, 6, 7, 3, POSITIVE); // Set the LCD I2C address

// Globale variabelen
unsigned long int milliSec = 0; // Opslag van de laatst gemeten millis() tijd
unsigned long int aanTijd = 0; // Tijd in mS dat het licht is aangegaan
int minuut = 0; // Laatste tijd in minuten
int uur = 0; // Laatste tijd in uren
int dag = 0; // Laatste tijd in dagen
int oudeDag = 0; // Dag-waarde uit de vorige keer loop()
int tempMax = 0; // Max. gemeten Temp, na de vorige spoelbeurt
String spoelBericht = "!! Niet gespoeld !!"; // Actuele geschiedenis van de spoelbeurt
String legeRegel = " "; // Voor de display-functies, 20x spatie

// de setup functie runt 1x bij reset of power-start
void setup() {
    Serial.begin(9600); // Alleen nodig voor het debuggen met de PC
    Serial.println("Debuggen van dit programma");

    // initialiseer digital pin 8-13 als output naar de 24V magnetische kleppen
    // dit werkt dan via de Velleman VMA05 shield op relais nr 1-6
    // in de kast heeft ieder relais een 2-punts kroonsteentje (maak/breek)
    int a = 8;
    while( a < 14 ) { pinMode( a, OUTPUT); a++; } // While loop executie

    pinMode(2, INPUT); // Geluidsensor op digital pin 2

    lcd.begin(20,4); // Initialiseer lcd voor 20 chars x 4 lines
    for(int i = 0; i < 3; i++) { // Knipper 3x met backlight v.d. 4x20 display
        lcd.backlight(); delay(250);
        lcd.noBacklight(); delay(250);
    }

    lcd.backlight(); spoelen(0); // voer de spoel actie uit, met backlight aan
    delay(500); // laatst spoel-info op 4x20 display
    lcd.setCursor(0,3); lcd.print(legeRegel); // Start maak line 3 leeg op 4x20 display
    lcd.setCursor(0,2); lcd.print(legeRegel); // Start maak line 2 leeg op 4x20 display
    lcd.setCursor(0,2); // Start bij character 0 op line 2
    lcd.print(spoelBericht); // Meld het laatste spoelBericht
    aanTijd = millis(); // aanTijd, na 20s display-licht uit doen
} // Einde setup ()

```



```

// de loop function runt oneindig vaak (=tot aan een crash of reset)
void loop() {

  milliSec = millis(); // Bewaar de stand v.d. milliSec tijdsteller
  if (simulatieAan) { milliSec = milliSec*1000; } // Simulatie: 1000x versnelt, 1 uur = 3.6 Sec
  oudeDag = dag; // Bewaar de dag-waarde van de vorige loop()
  minuut = milliSec/60000;
  uur = minuut/60;
  dag = uur/24; // Bereken de nieuwe dag, uur & minuut
  uur = uur - 24*(dag);
  minuut = minuut - 60*(uur+24*dag);

  lcd.setCursor(0,3); lcd.print(legeRegel); // Maak line 3 helemaal leeg op 4x20 display
  lcd.setCursor(0,3); // Start bij character 0 op line 3
  lcd.print(dag); lcd.print(" dag, "); // Nieuwe dag, uur & minuut op 4x20 display
  lcd.print(uur); lcd.print(" uur, ");
  lcd.print(minuut); lcd.print(" mn");

  temperaturen(0);

  if ((dag!=oudeDag)&&((tempMax>24)or(dag>6))) { // Criterium voor reset & spoelen
    lcd.print("Reset Arduino !!!"); // Reset de arduino vanuit de software
    { asm volatile (" jmp 0"); } // Dit is de arduino crash/reset-code
    lcd.print("Reset met Stekker!!!"); // Voor als de software reset zou falen
    exit (EXIT_FAILURE); // Stop en wacht op "uitschakelen" ...
  } // Na deze software reset begint de Arduino weer helemaal opnieuw op 0 mS

  if (geluidCheck(0)) { // bij geluid display-licht aan
    aanTijd = millis();
    lcd.backlight();
  } // geluidCheck functie duurt 500 ms voor stabiel display
  if ( millis()>(aanTijd+30*1000) ) { // schakel display-licht na 30 sec weer uit
    lcd.noBacklight();
  } // millis
} // Einde loop()

```

```

// de speel functie voor de kranen 1-6
void spoelen (int dummy) {
  lcd.setCursor(0,0); lcd.print(legeRegel);
  lcd.setCursor(0,0); // Start bij character 0 op line 0
  lcd.print("Legionella Preventie");
  if (simulatieAan) {
    lcd.setCursor(15,0); lcd.print(". SIM");
  }
  lcd.setCursor(0,1); lcd.print(legeRegel);
  lcd.setCursor(0,1); // Start bij character 0 on line 1
  lcd.print("Kranen check&spoelen");
  lcd.setCursor(0,2); lcd.print(legeRegel); // Start maak line 2 leeg
  lcd.setCursor(0,3); lcd.print(legeRegel); // Start maak line 3 leeg
  String bericht = "Gespoeld:";
  int x[6] = { 0, 7,14, 0, 7,14 }; // x-Posities op het display
  int y[6] = { 2, 2, 2, 3, 3, 3 }; // y-Posities op het display

  // Kraan/Relais 1 t/m 6 is Arduino kanaal 8-13 & i=0-5
  for(int i = 0; i<6; i++) {
    digitalWrite(i+8, HIGH); // LED bij dit relais aan (HIGH is v-level)
    if (stroomloopt(0)) {
      lcd.setCursor(x[i],y[i]); lcd.print("Spoel ");
      if (simulatieAan) { delay(1000); }
      else { delay(120000); } // 2 min: spoelen of 1 sec. wachten
      lcd.setCursor(x[i],y[i]); lcd.print(" ");
      lcd.print(i+1); lcd.print("-OK "); // Spoelen gelukt met dit relais
      if (bericht.length()<10) { bericht = bericht + " "; }
      else { bericht = bericht + "-"; }
      bericht = bericht + (i+1); // Status regel voor spoelen aanvullen
    }
    else {
      lcd.setCursor(x[i],y[i]);
      lcd.print(i+1); lcd.print("-Leeg"); // Dit relais is niet aangesloten
    }
    digitalWrite(i+8, LOW); // Zet de LED bij dit relai uit (LOW)
  } // for i loop

  delay(500);
  if (bericht.length()<10) { bericht = "Geen kraanrelais !!!"; }
  spoelBericht = bericht;
  lcd.setCursor(0,2); lcd.print(legeRegel);
  lcd.setCursor(0,2); // Start met character 0 op line 2
  lcd.print(spoelBericht);
  lcd.setCursor(0,3); lcd.print(legeRegel);
  lcd.setCursor(0,3); // Start met character 0 op line 3
} // Einde spoelen()

```

```

// In stroomloop de stroom meten en vergelijken met de detectie grens via een diode-combo
boolean stroomloopt (int dummy) {

    // De AC-stroom wordt gemeten als een DC-spanning met een combo van 1 + 3 diodes.
    // Deze zijn geplaatst in de stroomlijn van 24V-AC. Omdat 1 diode ca. -0.5V geeft
    // tijdens de negatieve stroom periode en 3 diodes ca +1.5V tijdens de positieve periode
    // daarmee wordt de gemiddelde spanning dan ca +1V over dit diode-combo de spanning
    // wordt ca 25x gemeten met 2ms interval en opgeteld in bytes de grenswaarde
    // is bepaald op 3000, met het debug statement op de seriele poort.
    delay(500);
    int sensorReading = 0;
    for(int i = 0; i< 25; i++) {
        delay(2); sensorReading = sensorReading + analogRead(0);
    }
    // Serial.println(sensorReading); // Serial print voor debuggen
    if (sensorReading>4000) return true; // Er is wel stroom gemeten (>4000)
    else return false; // Er is geen stroom gemeten
} // Einde stroomloopt()

// Meet alle temperaturen en corrigeer de Tmax-waarde.
void temperaturen (int dummy) {

    int sensorValue = 0; // Gemeten waarde in bits 0 t/m/ 1024
    float milliVolt = 0.0; // Gemeten waarde in mV
    int temp = 0; // Gemeten Temperatuur (=intern)
    int tempExt = 0; // Gemeten Temp-extern

    for(int i = 0; i< 20; i++) { // Doe 20 metingen met 1 mS delay
        delay(1);
        sensorValue = sensorValue + analogRead(1); // Temp sensor Arduino #1 = VMA05 #2
    }
    temp = sensorValue*25.0/(1024); // geldt voor 20 metingen opgeteld

    delay(10); sensorValue = 0;
    for(int i = 0; i< 20; i++) { // Doe 20 metingen met 1 mS delay
        delay(1);
        sensorValue = sensorValue + analogRead(2); // T-ext sensor Arduino #2 = VMA05 #3
    }
    tempExt = sensorValue*25.0/(1024); // geldt voor 100 metingen opgeteld

    if (temp > tempMax) { tempMax=temp; };
    if (tempExt > tempMax) tempMax=tempExt;

    lcd.setCursor(0,1); lcd.print(legeRegel);
    lcd.setCursor(0,1); // Start met character 0 op line 1
    lcd.print("T=");
    if (temp>0) { lcd.print(temp); }
    else { lcd.print("-X"); } // Print "-X" als de T kleiner dan 1 is
    lcd.print(" T_x=");
    if (tempExt>0) { lcd.print(tempExt); }
    else { lcd.print("-X"); }
    lcd.print(" T_max="); lcd.print(tempMax); // Print "-X" als de T-ext kleiner dan 1 is
} // einde temperaturen

// Controleer voor 500 ms of de geluidsensor iets waarneemt (Arduino Dig-in 2 = VMA Dig-in 1)
boolean geluidCheck (int dummy) {

    // De grenswaarde instellen met de instel potmeter op de kleine geluidsensor-print
    // Stilte geeft een 1 op digitalRead(2) --> false
    // Geluid geeft een 0 op digitalRead(2) --> true
    unsigned long int stopTimer;
    boolean check = false;
    stopTimer = millis()+500;
    while ( millis() < stopTimer ) {
        check = digitalRead(2);
        if (check==0) { return true; }
    } // while
    if (check==1) { return false;}
} // Einde geluidCheck()

// Einde van het LegioSpoel arduino programma.

```